
Chaos_In_OTA_Filters
 *   dsauersanjose@aol.com 
 *   www.idea2ic.com                               R3 = 1/A0
 *   4/15/08             ________________________/\  /\  /\__
 *                      |                          \/  \/    |
 *                      |               R2 = 1/A1            |
 *                      |____________/\  /\  /\_             |
 *                      |              \/  \/   |            |
 *                      |   R1 = 1              |BP          |LP
 *                      |_/\  /\  /\_ HP        |            |
 *          R0 = 1/A0   |   \/  \/   |          |            |
 *     ___              | |\         | XOTA_S1  |  XOTA_S2   |
 *    |IN |_/\  /\  /\__|_|-\        | _______  |  _______   |  ___
 *    |___|   \/  \/  SUMN|  \_______|_|      |_|__|      |__|_|OUT|
 *     _|_                |  /         | 1/s  |    | 1/s  |    |___|
 *    /VIN\             __|+/ OPA1     |______|    |______|
 *    \___/            |  |/
 *      |             _|_        H(s) = 1/(s^2 +s*A1   + A0)
 *     _|_            ///        H(s) = 1/(s^2 +s*w0/Q + w0^2)
 *     ///
 *             Set  A0 = 1 and scale s to 1KHz
 *             Then R2 = Q and  s = 2*PI*1KHz

.OPTIONS  GMIN=1p     METHOD=euler  ABSTOL=1n  TEMP=27 srcsteps = 100  gminsteps = 10 ITL1=400  

.OPTIONS RELTOL=.001  ABSTOL=1p     VNTOL=1p      ITL4=500
*======== ====== ====== ====== ====== ====== ====== ====== ======
VT        Vtime  0      PWL    ( 0 0 1 1 ) 
VF        VF     0      DC     1k
VF2       VF2    0      PWL    ( 0 2K 1 0 )
V_IN      VIN    0      DC     0     SIN( 0 .1 .7k 1p )
*BIN      BIN    0      v =    1.2*v(Vtime)*(sin(6.28319*v(VF)*v(Vtime))+.3*sin(6.28319*v(VF)*v(Vtime)*2))
BIN       BIN    0      v =    0.2*v(Vtime)*(sin(6.28319*v(VF)*v(Vtime)))
R0        BIN    SUMN   10k    
R1        SUMN   HP     10k
R2        SUMN   BP     100k
R3        SUMN   LP     10k
XOPA1     SUMN   0      HP    OPA
XOTAS1    HP     BP           OTA_S
XOTAS2    BP     LP           OTA_S

*==OTAs_Can_Perform_The_Exact_Same_Function====
.control
tran    .1m     .01      0 
plot   bp  title   StateVariable_Q_10
plot bp vs bin
plot lp vs bin
plot hp vs bin
.endc
  
.SUBCKT OTA_S IN      OUT 
QN1         VC1      IN     VE1    NPNP
QN2         VC2      0      VE1    NPNP 
QP1         VC1      VC1    VCC    PNPP
QP2         VC2      VC1    VCC    PNPP
IB         VE1      0             5.2u
VCC         VCC      0             DC     2
EGN2        OUT      0      VC2    0      +1
C1          VC2      0            .01592u
.ends
*          ________________
*        _|_   |  VCC      |
*       /VCC\  |           |   C1
*       \___/   ->       <-    _||_            ___
*        _|_   QP1`|___|'QP2  | ||_|_    _____|OUT|
*        ///    _ '| | |`_    |   ///   _|_   |___|     .SUBCKT   OTA_S  IN   OUT
*              |_____| VC2|___|_______ /   \            ___  _______    ___
*    ___      _| VC1      |_          | Egn2|   ---->  |IN |_|      |__|OUT|
*   |IN |___|'QN1       QN2 `|___    _|  +1 |   ---->  |___| | 1/s  |  |___|
*   |___|   |`->  VE1     <-'|  |   |  \___/                 |______|
*              |__________|    _|_  |____|
*                   |          ///      _|_            s  = jw0    s = 2*PI*Freq
*                  _|_                  ///            @s = jw0    R = 1/sC
*                 / _ \IB
*                 \/ \/
*                 /\_/\        IOUT = IB*TANH( VIN/52mV) =>VIN*IB/52mV
*                 \___/
*                   |
*                  _|_
*                  ///



.SUBCKT OPA   INP    INN    OUT
EGN1        O1     0      INP    INN    -1
EGN2        OUT    0      IP     0      -1000000
R1          O1     IP     10k 
C1          OUT    IP     3p
.ends
*
*  http://www.idea2ic.com/PlayWithJavascript/R_C_Freq.html
*
*    ___         O1         C1 10pF   ___     .SUBCKT  OPA INP INN OUT
*   |INP|_     _____        _||______|OUT|    
*   |___| |   _|_   |   R1 | || _|_  |___|     ___   |\
*         |_ /   \  |_/\  _|__ /   \          |INP|__|-\    ___
*    ___    | EGN1|  10K\/  IP| EGN2|  ---->  |___|  |  \__|OUT|
*   |INN|___|  -1 |          _| 90dB|  ---->   ___   |  /  |___|
*   |___|    \___/       IP |  \___/          |INN|__|+/
*              |            |____|            |___|  |/ OPA
*             _|_               _|_
*             ///               ///

.model    NPNP   NPN(   BF=2100 VAF=216 )

.model    PNPP   PNP(   BF=2100 VAF=210)

.end

=====================END_OF_SPICE=======================

The methods used to observe chaos may provide
a better way to view a filter as to whether
is operating properly. 

The waveform above is at the exact resonance
frequency of a two pole LM3080 type filter. 
The input signal is ramped up from 0 to 20mV. 



Using the chaos technique of plotting Output
versus input, the plot at the Bandpass output
is a straight line similar to a resistor.
As the input signal increases, the line
essentially widens.  

But the LowPass and HighPass outputs are
phase shifted by 90degrees. As the input
signal increases in magnitude, the shapes
at all output increase in size but don't
change in shape. 



Since the LowPass and HighPass are at different
phases, their rotations are opposite. 

Now when the input signal is made larger,
the output at the Bandpass is beginning to
act a little different. 

But the plotting of the Bandpass output 
versus the input makes it easy to see when
the transfer function is no longer a straight



line. This looks like a good way to test
whether or not a filter is behaving properly.

The wave forms at both the LowPass and HighPass
output also change shape. What is happening
is that the inputs to the OTA are distorting
enough that the assumption of linearity are
obviously no longer valid.

This perhaps answers a question of how low
must the THD be in an OTA based filter?
The answers is when does the filter begin
to effectively operate in another mode?



When the input signal is increased further,
now we are begining to generate chaos curves. 

Some key elements to a Chaos circuit appear
to be feedback and nonlinearity. 

The nonlinear feedback may allow input signal



to be interpedated in more than one way.

Perhaps the multiple states can result from
the output distorted signal having multiple
ways to regenerate its own input signal. 


